The Mathematica® Journal

Tricks of the Trade

Edited by Paul Abbott

This is a column of programming tricks and techniques, most of which, we hope, will be contributed by our readers, either directly as submissions to *The Mathematica Journal* or as an edited answer to a question posted in the *Mathematica* newsgroup, comp.soft-sys.math.mathematica.

■ Sum-Free Set

The *sumset* of two or more subsets of an additive group is the set of all sums formed by taking one element from each set (see planetmath.org/sumset.html). The sumset can be computed using **Tuples**.

```
SumSet[s__List] := Union[Total /@ Tuples[{s}]]
```

Define ⊕ to be `SumSet`.

```
CirclePlus := SumSet
```

Here is the sumset $\{1, 2\} \oplus \{1, 3, 5\} \oplus \{2\}$.

```
{1, 2} ⊕ {1, 3, 5} ⊕ {2}
```

{4, 5, 6, 7, 8, 9}

A sum-free set S is a set for which the intersection of S and the sumset $S \oplus S$ is empty (see mathworld.wolfram.com/Sum-FreeSet.html).

```
SumFreeQ[s_List] := s ⋂ (s ⊕ s) == {}
```


For example, the sum-free subsets of {1, 2, 3} are ∅ ≡ {}, {1}, {2}, {3}, {1, 3}, and {2, 3}.

```
SumFreeQ /@ {{}, {1}, {2}, {3}, {1, 3}, {2, 3}}
```

{True, True, True, True, True, True}

Note that {1, 2} is not sum-free.

```
SumFreeQ[{1, 2}]
```

False

Here are the sum-free subsets of {1, 3, 5, 7, 8}.

```
Select[Subsets[{1, 3, 5, 7, 8}], SumFreeQ]
```

{{}, {1}, {3}, {5}, {7}, {8}, {1, 3}, {1, 5}, {1, 7}, {1, 8}, {3, 5},
{3, 7}, {3, 8}, {5, 7}, {5, 8}, {7, 8}, {1, 3, 5}, {1, 3, 7}, {1, 3, 8},
{1, 5, 7}, {1, 5, 8}, {3, 5, 7}, {3, 7, 8}, {5, 7, 8}, {1, 3, 5, 7}}

Sum-free subsets of {1, 2, … , *n*} can be computed recursively as follows.

```
SumFreeSet[0] = {{}};

SumFreeSet[n_] :=
  SumFreeSet[n] =
    SumFreeSet[n - 1] ⋃
      (# ⋃ {n} &) /@ Select[SumFreeSet[n - 1], # ⋂ (n - #) == {} &]
```

The key to this computation is the use of the test **# ⋂ (*n* − #) == {} &** on **SumFreeSet[*n* − 1]** to construct elements of **SumFreeSet[*n*]**.

Here are the sum-free subsets for *n* = 0, 1, … , 4.

```
Column[Table[SumFreeSet[n], {n, 0, 4}]]
```

{{}}
{{}, {1}}
{{}, {1}, {2}}
{{}, {1}, {2}, {3}, {1, 3}, {2, 3}}
{{}, {1}, {2}, {3}, {4}, {1, 3}, {1, 4}, {2, 3}, {3, 4}}

Alternatively, sum-free subsets can be computed using **NestList**, starting from the empty set.

```
Module[{n = 0},
 Column[
  NestList[
   (++n; #1 ⋃ (#1 ⋃ {n} &) /@ Select[#1, #1 ⋂ n - #1 == {} &]) &,
   {{}}, 5]]]
```

{{}}
{{}, {1}}
{{}, {1}, {2}}
{{}, {1}, {2}, {3}, {1, 3}, {2, 3}}
{{}, {1}, {2}, {3}, {4}, {1, 3}, {1, 4}, {2, 3}, {3, 4}}
{{}, {1}, {2}, {3}, {4}, {5}, {1, 3}, {1, 4}, {1, 5},
{2, 3}, {2, 5}, {3, 4}, {3, 5}, {4, 5}, {1, 3, 5}, {3, 4, 5}}

The number of sum-free subsets for each n are 1, 2, 3, 6, 9, 16, …. Searching for this sequence at oeis.org, we find that it is A007865.

```
Length /@ First[%]
```

{1, 2, 3, 6, 9, 16}

Using **Sow** and **Reap**, here is the number of sum-free subsets for $0 \leq n \leq 25$.

```
Module[{n = 0},
  First @
    Last @
      Reap @
        Nest[
          (++n; Sow[Length[#]];
              # ⋃ (# ⋃ {n} &) /@ Select[#, # ⋂ (n - #) == {} &]) &, {{}}, 25]]
```

{1, 2, 3, 6, 9, 16, 24, 42, 61, 108, 151, 253, 369, 607, 847,
1400, 1954, 3139, 4398, 6976, 9583, 15 456, 20 982, 32 816, 45 417}

Asymptotic Expansion and π

Gregory's series (mathworld.wolfram.com/GregorySeries.html) is a slowly convergent formula for π.

$$4 \sum_{k=1}^{\infty} \frac{(-1)^{k-1}}{2k-1}$$

π

Truncating the series after 50,000 terms (half a power of 10, in this case 10^5) yields a result that is incorrect in the 6^{th} digit.

$$\text{gregory} = 4\grave{}50 \sum_{k=1}^{50\,000} \frac{(-1)^{k-1}}{2k-1}$$

3.1415726535897952384626423832795041041971666293751

pi = *N*[π, 50]

3.1415926535897932384626433832795028841971693993751

$$1 - \lfloor \log_{10}(\text{pi} - \text{gregory}) \rfloor$$

6

However, comparing these two numbers, it is surprising how many digits they have in common [1, 2].

RealDigits[pi] − RealDigits[gregory]

{{0, 0, 0, 0, 0, 2, 0, 0, 0, 0, 0, 0, 0, 0, 0, −2, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0,
0, 0, 0, 0, 0, 0, 0, −2, 7, 8, 0, 0, 0, 0, 0, 0, 0, 3, −3, 7, 0, 0, 0, 0, 0}, 0}

Moreover, the index of the position of the least significant digit of each block of different digits is an odd multiple of 5.

Partition[Rest@First[%], 5]

$$\begin{pmatrix} 0 & 0 & 0 & 0 & 2 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -2 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -2 & 7 & 8 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 3 & -3 & 7 & 0 \end{pmatrix}$$

The differences can be computed using **FromDigits**.

FromDigits /@ %

{2, 0, −2, 0, 10, 0, −122, 0, 2770}

We can represent the difference between π and Gregory's series truncated after 50,000 terms as

```
                2                      122
3.1415926535897932384626433832795028841971693993751,
      2                  10                    2770
```

where numbers above the center line are negative and those below the line are positive.

Searching for the sequence of differences at The On-Line Encyclopedia of Integer Sequences™ (OEIS™) , we find that they are twice the Euler numbers, E_{2k} (oeis.org/A011248).

Table[{*k*, 2 E_{2k}}, {*k*, 0, 4}]

$$\begin{pmatrix} 0 & 2 \\ 1 & -2 \\ 2 & 10 \\ 3 & -122 \\ 4 & 2770 \end{pmatrix}$$

Empirically, we have determined the asymptotic difference between π and the truncated Gregory's series.

$$\textbf{diff[n\_, m\_]} = 2\sum_{k=0}^{m} \frac{E_{2k}}{n^{2k+1}};$$

See [1] for a proof of this result.

Adding the asymptotic difference to the truncated Gregory's series and putting $n = 2 \times 50\,000 = 10^5$, we can recover π to (at least) 50 decimal places.

$$\pi - \textbf{gregory} - \textbf{diff}\left[10^5, 4\right]$$

$$0. \times 10^{-50}$$

The asymptotic difference can be computed directly using **Series**.

$$\textbf{Assuming}\left[n > 0 \bigwedge \frac{n}{4} \in \mathbb{Z},\right.$$

$$\left.\textbf{FullSimplify}\left[\textbf{Series}\left[\textbf{FunctionExpand}\left[\pi - 4\sum_{k=1}^{\frac{n}{2}} \frac{(-1)^{k-1}}{2k-1}\right], \{n, \infty, 9\}\right]\right]\right]$$

$$\frac{2}{n} - \frac{2}{n^3} + \frac{10}{n^5} - \frac{122}{n^7} + \frac{2770}{n^9} + O\left(\left(\frac{1}{n}\right)^{10}\right)$$

See also [3].

Hadamard Regularization

Hadamard regularization is a technique for handling divergent integrals (essentially keeping only the finite part of the integral) and plays an important role in several branches of mathematical physics (see [4, 5] and mathworld.wolfram.com/HadamardIntegral.html). Consider evaluating

$$K^{\alpha}[f] \equiv \frac{1}{\Gamma(-\alpha)} \int_{-1}^{1} \frac{f(y)}{(1-y)^{\alpha+1}}\, dy$$

in the Hadamard sense, where $0 \leq n < \alpha < n+1$ and $n \in \mathbb{Z}$, that is, $n = \lfloor \alpha \rfloor$ and $f \in C^{n+1}[-1, 1]$.

Using integration by parts via pattern matching, we can increase the exponent of $(1-y)^{-p}$ until it is integrable, that is, $-1 < p \leq 0$.

$$\textbf{byparts} = \int (1-y)^{\text{p\_}}\, \text{f\_}[y]\, dy \mapsto f(y) \int (1-y)^p\, dy - \int \left(\int (1-y)^p\, dy \right) \partial_y f(y)\, dy;$$

Here is the formal result of integrating by parts once.

$$\frac{1}{\Gamma(-\alpha)} \int \frac{f(y)}{(1-y)^{\alpha+1}}\, dy \text{ /. byparts // FullSimplify // Expand}$$

$$\frac{f(y)\,(1-y)^{-\alpha}}{\alpha\,\Gamma(-\alpha)} - \frac{\int (1-y)^{-\alpha} f'(y)\, dy}{\alpha\,\Gamma(-\alpha)}$$

The $(1-y)^{-\alpha} f(y)$ term is singular at $y = 1$ if $\alpha > 0$. Here is the result of three partial integrations.

$$\text{Collect}\left[\text{Nest}\left[\# \text{ /. byparts \&}, \frac{1}{\Gamma(-\alpha)} \int \frac{f(y)}{(1-y)^{\alpha+1}}\, dy, 3\right], \{f(y), f^{(\_)}(y)\},\ \text{FullSimplify}\right]$$

$$\frac{\int f^{(3)}(y)\,(1-y)^{2-\alpha}\, dy}{\Gamma(3-\alpha)} - \frac{(1-y)^{2-\alpha} f''(y)}{\Gamma(3-\alpha)} - \frac{(1-y)^{1-\alpha} f'(y)}{\Gamma(2-\alpha)} - \frac{f(y)\,(1-y)^{-\alpha}}{\Gamma(1-\alpha)}$$

Neglecting the singular terms at $y = 1$, we evaluate the partial integrals at $y = -1$.

$$-\% \text{ /. HoldPattern[Integrate[\_\_]] :> 0 /. } y \to -1$$

$$\frac{2^{2-\alpha} f''(-1)}{\Gamma(3-\alpha)} + \frac{2^{1-\alpha} f'(-1)}{\Gamma(2-\alpha)} + \frac{2^{-\alpha} f(-1)}{\Gamma(1-\alpha)}$$

The pattern is clear. Dropping the singular terms at $y = 1$, we obtain

$$K^{\alpha}[f] \equiv \frac{1}{\Gamma(-\alpha)} \int_{-1}^{1} \frac{f(y)}{(1-y)^{\alpha+1}}\, dy =$$

$$\sum_{k=0}^{n} \frac{2^{k-\alpha}}{\Gamma(k-\alpha+1)} f^{(k)}(-1) + \frac{1}{\Gamma(n-\alpha+1)} \int_{-1}^{1} (1-y)^{n-\alpha} f^{(n+1)}(y)\, dy.$$

As a definite example, consider

$$\int_{-1}^{1} \frac{\exp(y)}{(1-y)^{\alpha+1}}\, dy.$$

Direct integration followed by series expansion about $\epsilon = 0$ reveals the singular terms.

$$\textbf{Assuming}\left[1 > \epsilon > 0,\ \frac{1}{\Gamma(-\alpha)} \int_{-1}^{1-\epsilon} \frac{\exp(y)}{(1-y)^{\alpha+1}}\, dy\right]$$

$$\frac{e\,(\Gamma(-\alpha,\epsilon) - \Gamma(-\alpha,2))}{\Gamma(-\alpha)}$$

Series[%, {ϵ, 0, 1}] // ExpandAll

$$-\frac{e\,\Gamma(-\alpha,2)}{\Gamma(-\alpha)} + \epsilon^{-\alpha}\left(\frac{e}{\alpha\,\Gamma(-\alpha)} - \frac{e\,\epsilon}{(\alpha-1)\,\Gamma(-\alpha)} + O(\epsilon^2)\right) + e$$

Now $\epsilon^{-\alpha}$ is singular at $\epsilon = 0$ for $\alpha > 0$, and $\epsilon^{1-\alpha}$ is either singular if $\alpha > 1$ or vanishes if $0 < \alpha < 1$. So both terms are ignorable. Hence the nonsingular part can be extracted as follows.

$$K_{\alpha\_}[\textbf{Exp}] = \%\ /.\ \epsilon^{-\alpha} \to 0\ //\ \textbf{FullSimplify}$$

$$e - \frac{e\,\Gamma(-\alpha,2)}{\Gamma(-\alpha)}$$

For example, here is the exact result for $\alpha = 3/2$.

$$K_{3/2}[\textbf{Exp}]\ //\ \textbf{FunctionExpand}\ //\ \textbf{Simplify}$$

$$e\,\mathrm{erf}\left(\sqrt{2}\right) + \frac{3}{4\,e\,\sqrt{2\pi}}$$

Alternatively, using the identity obtained using integration by parts, we obtain the same answer.

$$\textbf{Module}\Big[\{\alpha = 3/2,\ n,\ f = \textbf{Exp}\},\ n = \textbf{Floor}[\alpha];$$
$$\sum_{k=0}^{n} \frac{2^{k-\alpha}}{\Gamma(k-\alpha+1)} f^{(k)}(-1) + \frac{1}{\Gamma(n-\alpha+1)} \int_{-1}^{1} (1-y)^{n-\alpha} f^{(n+1)}(y)\, dy\Big]\ //$$
Simplify

$$e\,\mathrm{erf}\left(\sqrt{2}\right) + \frac{3}{4\,e\,\sqrt{2\pi}}$$

■ References

[1] J. M. Borwein, P. B. Borwein, and K. Dilcher, "Pi, Euler Numbers, and Asymptotic Expansions," *American Mathematical Monthly*, **96**(8), 1989 pp. 681–687.

[2] G. Almkvist, "Many Correct Digits of π, Revisited," *American Mathematical Monthly*, **104**(4), 1997 pp. 351–353. DOI-Link: dx.doi.org/10.2307/2974583

[3] S. Matsumoto, "Convergence Improvement of Infinite Series by Linear Fractions," in *Applied Mathematica: Electronic Proceedings of the Eighth International Mathematica Symposium (IMS06)*, Avignon, France (Y. Papegay, ed.), Rocquencourt: INRIA, 2006 ISBN 2-7261-1289-7.

[4] D. Elliott, "Three Algorithms for Hadamard Finite-Part Integrals and Fractional Derivatives," *Journal of Computational and Applied Mathematics*, **62**(3), 1995 pp. 267–283.

[5] L. Blanchet and G. Faye, "Hadamard Regularization," *Journal of Mathematical Physics*, **41**, 2000 pp. 7675–7714.

P. Abbott, "Tricks of the Trade," *The Mathematica Journal*, 2012. dx.doi.org/10.3888/tmj.10.3-1.

About the Author

Paul Abbott
School of Physics, M013
The University of Western Australia
35 Stirling Highway
Crawley WA 6009, Australia
tmj@physics.uwa.edu.au
physics.uwa.edu.au/~paul